National and Kapodistrian University of Athens (NKUA)
MSc in Data Science and Information Technologies (DSIT)
Course: *Knowledge Technologies (2025–2026)*

Student: *Konstantinos Kritsotakis*
Registration Number: *7115152500007*
Background: *Chemical Engineer, National Technical University of Athens (NTUA)*
Graduate Student: *MSc DSIT – Specialization: Big Data & Artificial Intelligence*

A)

1.

Query

```
PREFIX kg:   <http://www.ai.di.uoa.gr/pokemon/>
PREFIX geof: <http://www.opengis.net/def/function/geosparql/>

SELECT DISTINCT ?person
WHERE {
  ?person a kg:Person ;
          kg:hasLocation ?pWKT .

  kg:CentricoPlaza kg:hasLocation ?plazaWKT .

  FILTER( geof:sfWithin(?pWKT, ?plazaWKT) )
}
```

Results

person
http://www.ai.di.uoa.gr/pokemon/CalebCedar_342
http://www.ai.di.uoa.gr/pokemon/HanaWillow_536
http://www.ai.di.uoa.gr/pokemon/TaliaDriftLeader_16
http://www.ai.di.uoa.gr/pokemon/NoraDrift_665
http://www.ai.di.uoa.gr/pokemon/GioHawke_442
http://www.ai.di.uoa.gr/pokemon/MinaSkylark_396
http://www.ai.di.uoa.gr/pokemon/HanaPine_251
http://www.ai.di.uoa.gr/pokemon/JudeSilver_962
http://www.ai.di.uoa.gr/pokemon/SageKeen_250
http://www.ai.di.uoa.gr/pokemon/PanosCosmatos_1296
http://www.ai.di.uoa.gr/pokemon/BruceStevenson_1990
http://www.ai.di.uoa.gr/pokemon/ReiSpark_994

```
http://www.ai.di.uoa.gr/pokemon/NoraVale_126
http://www.ai.di.uoa.gr/pokemon/StelaFigwood_1259
http://www.ai.di.uoa.gr/pokemon/BellaWoodworth_2105
```

2.

Query

```
PREFIX kg:   <http://www.ai.di.uoa.gr/pokemon/>
PREFIX geof: <http://www.opengis.net/def/function/geosparql/>

SELECT DISTINCT ?person ?info
WHERE {
  ?person a kg:Person ;
        kg:hasLocation ?pWKT ;
        kg:hasInformation ?info .

  kg:CentricoPlaza kg:hasLocation ?plazaWKT .

  FILTER( geof:sfWithin(?pWKT, ?plazaWKT) )
}
```

Results

CalebCedar_342	This person has spent many years living alongside PokГ©mon and supporting trainers and travelers.
HanaWillow_536	This person is a familiar face in the PokГ©mon world and often helps others on their journeys.
TaliaDriftLeader_16	This person has spent many years living alongside PokГ©mon and supporting trainers and travelers.
NoraDrift_665	A respected member of the community, this person understands the bond between humans and PokГ©mon.
GioHawke_442	Known throughout nearby regions, this individual plays an active role in everyday PokГ©mon society.
MinaSkylark_396	This person has spent many years living alongside PokГ©mon and supporting trainers and travelers.
HanaPine_251	A respected member of the community, this person understands the bond between humans and PokГ©mon.
JudeSilver_962	This individual contributes to the PokГ©mon world through daily work and close interaction with PokГ©mon.
SageKeen_250	A respected member of the community, this person understands the bond between humans and PokГ©mon.
PanosCosmatos_1296	I just direct movies?
BruceStevenson_1990	*Whistles*
ReiSpark_994	This person has spent many years living alongside PokГ©mon and supporting trainers and travelers.
NoraVale_126	This person is a familiar face in the PokГ©mon world and often helps others on their journeys.
StelaFigwood_1259	I am sorry deart, I haven't seen anything like that...
BellaWoodworth_2105	Oh my! Your Pikachu is missing, how terrible! You could ask for help in <http://www.ai.di.uoa.gr/pokemon/PoliceStation_LumioseCity>!

From the info strings, the URI of the police station that will be used in the next query is:
<http://www.ai.di.uoa.gr/pokemon/PoliceStation_LumioseCity\>.”

B)

1.

Query

```
PREFIX kg:   <http://www.ai.di.uoa.gr/pokemon/>
PREFIX geof: <http://www.opengis.net/def/function/geosparql/>

SELECT DISTINCT ?entity ?name
WHERE {
 # entity = Person OR Pokemon
 { ?entity a kg:Person . }
 UNION
 { ?entity a kg:Pokemon . }

 # name
 ?entity kg:hasName ?name .

 # spatial presence: entity location within police station polygon
 ?entity kg:hasLocation ?eWKT .
 kg:PoliceStation_LumioseCity kg:hasLocation ?psWKT .

 FILTER( geof:sfWithin(?eWKT, ?psWKT) )
}
ORDER BY LCASE(STR(?name))
```

Results

entity	name
Bulbasaur_75	Bulbasaur
Dratini_14	Dratini
Officer_ErikaBloom	Erika Bloom
Flareon_78	Flareon
Golbat_79	Golbat
Officer_KogaNightwind	Koga Nightwind
Officer_LtSurgeVoltman	Lt. Surge Voltman
OfficerJenny	Officer Jenny
Poliwrath_68	Poliwrath
Vileplume_79	Vileplume

2.

Query

```
PREFIX kg:   <http://www.ai.di.uoa.gr/pokemon/>
PREFIX geof: <http://www.opengis.net/def/function/geosparql/>

SELECT DISTINCT ?officer ?name ?dist
WHERE {
 ?officer a kg:Person ;
         kg:hasName ?name ;
         kg:hasLocation ?oWKT ;
         kg:previousPost ?prevWKT .

 kg:PoliceStation_LumioseCity kg:hasLocation ?psWKT .
 FILTER( geof:sfWithin(?oWKT, ?psWKT) )

 kg:PalletTown kg:hasLocation ?palletWKT .

 BIND( geof:distance(?prevWKT, ?palletWKT) AS ?dist )
}
ORDER BY ?dist
LIMIT 1
```

Results

	officer	name	dist
1	kg:OfficerJenny	"Officer Jenny"	"37.09991303161328"^^xsd:double

Footnote:
In the provided dataset, the property kg:previousPost is represented as a point geometry (POINT) rather than a polygon or town entity. Therefore, topological relations such as borders between towns cannot be applied, and only point-to-polygon distance–based spatial reasoning is feasible.

C)

1.

Query

```
PREFIX kg:   <http://www.ai.di.uoa.gr/pokemon/>
PREFIX geof: <http://www.opengis.net/def/function/geosparql/>

SELECT ?parking ?dist
WHERE {
 ?parking a kg:Parking ;
         kg:hasLocation ?pWKT .

 kg:CentricoPlaza kg:hasLocation ?plazaWKT .

 BIND(
  geof:distance(
   ?pWKT,
```

```
    ?plazaWKT
   ) AS ?dist
  )
}
ORDER BY ?dist
LIMIT 5
```

Results

	parking	dist
1	kg:Parking_LumioseCity_21	"10.709302801421568"^^xsd:double
2	kg:Parking_LumioseCity_68	"11.469800909559956"^^xsd:double
3	kg:Parking_LumioseCity_108	"14.943818487912418"^^xsd:double
4	kg:Parking_LumioseCity_101	"16.385930806131206"^^xsd:double
5	kg:Parking_LumioseCity_100	"18.903223554447322"^^xsd:double

2.

Query

```
PREFIX kg:   <http://www.ai.di.uoa.gr/pokemon/>
PREFIX geof: <http://www.opengis.net/def/function/geosparql/>

SELECT DISTINCT ?clue ?clueWKT ?parking ?parkingWKT
WHERE {
 ?clue a kg:Clue ;
     kg:hasLocation ?clueWKT .

 VALUES ?parking {
  kg:Parking_LumioseCity_21
  kg:Parking_LumioseCity_68
  kg:Parking_LumioseCity_108
  kg:Parking_LumioseCity_101
  kg:Parking_LumioseCity_100
 }

 ?parking kg:hasLocation ?parkingWKT .

 FILTER( geof:sfWithin(?clueWKT, ?parkingWKT) )
}
ORDER BY ?parking ?clue
```

Results

	clue	clueWKT	parking	parkingWKT
1	kg:Clue_05	"POINT (394.1251617741508 185.1684889542749)"^^<http://www.opengis.net/ont/geosparql#wktLiteral>	kg:Parking_LumioseCity_100	"POLYGON ((400.95800172457325 174.56028074617174, 385.1978246317435 182.8208843920269, 385.5626663159592 184.85017283350217, 399.8689372419413 186.3672989060587, 400.95800172457325 174.56028074617174))"^^<http://www.opengis.net/ont/geosparql#wktLiteral>
2	kg:Clue_08	"POINT (336.37604900384645 216.40136616610704)"^^<http://www.opengis.net/ont/geosparql#wktLiteral>	kg:Parking_LumioseCity_101	"POLYGON ((336.0268182753267 209.4222394888424, 335.9967620870416 217.0802761336244, 338.8291017821888 224.03615618610453, 344.0572080215613 220.3278864729608, 351.23703869655014 213.37030659960453, 336.0268182753267 209.4222394888424))"^^<http://www.opengis.net/ont/geosparql#wktLiteral>
3	kg:Clue_06	"POINT (377.77028214436825 175.8169765760254)"^^<http://www.opengis.net/ont/geosparql#wktLiteral>	kg:Parking_LumioseCity_108	"POLYGON ((377.703010326816 174.5376565516479, 376.68192741214864 174.20629993827137, 375.8388936333329 175.8428562645463, 377.85724313545256 176.0966510907535, 377.92225192929953 175.55114015767944, 377.703010326816 174.5376565516479))"^^<http://www.opengis.net/ont/geosparql#wktLiteral>
4	kg:Clue_07	"POINT (377.4951211392621 175.3437634880398)"^^<http://www.opengis.net/ont/geosparql#wktLiteral>	kg:Parking_LumioseCity_108	"POLYGON ((377.703010326816 174.5376565516479, 376.68192741214864 174.20629993827137, 375.8388936333329 175.8428562645463, 377.85724313545256 176.0966510907535, 377.92225192929953 175.55114015767944, 377.703010326816 174.5376565516479))"^^<http://www.opengis.net/ont/geosparql#wktLiteral>
5	kg:Clue_01	"POINT (384.7877387928731 182.87441603760254)"^^<http://www.opengis.net/ont/geosparql#wktLiteral>	kg:Parking_LumioseCity_21	"POLYGON ((380.4228509911085 176.16431383682817, 376.46062110965346 183.00739674732844, 379.81939696694485 189.2600769215616, 382.27167670568696 192.74894390225745, 388.25739483991214 190.1869521694527, 394.5269161099987 184.0141956796327, 392.95848391629465 183.03065066851732, 380.4228509911085 176.16431383682817))"^^<http://www.opengis.net/ont/geosparql#wktLiteral>
6	kg:Clue_04	"POINT (386.36765443814807 188.83797202074777)"^^<http://www.opengis.net/ont/geosparql#wktLiteral>	kg:Parking_LumioseCity_21	"POLYGON ((380.4228509911085 176.16431383682817, 376.46062110965346 183.00739674732844, 379.81939696694485 189.2600769215616, 382.27167670568696 192.74894390225745, 388.25739483991214 190.1869521694527, 394.5269161099987 184.0141956796327, 392.95848391629465 183.03065066851732, 380.4228509911085 176.16431383682817))"^^<http://www.opengis.net/ont/geosparql#wktLiteral>
7	kg:Clue_02	"POINT (376.5559261176785 211.78233119289663)"^^<http://www.opengis.net/ont/geosparql#wktLiteral>	kg:Parking_LumioseCity_68	"POLYGON ((386.2380121687817 207.5780813935397, 386.4322413815465 202.61630433497015, 375.44164361279786 202.07786128026584, 369.8615118820355 224.1679636349119, 386.2380121687817 207.5780813935397))"^^<http://www.opengis.net/ont/geosparql#wktLiteral>
8	kg:Clue_03	"POINT (373.0031765149197 219.39165786666)"^^<http://www.opengis.net/ont/geosparql#wktLiteral>	kg:Parking_LumioseCity_68	"POLYGON ((386.2380121687817 207.5780813935397, 386.4322413815465 202.61630433497015, 375.44164361279786 202.07786128026584, 369.8615118820355 224.1679636349119, 386.2380121687817 207.5780813935397))"^^<http://www.opengis.net/ont/geosparql#wktLiteral>
9	kg:Clue_09	"POINT (383.2962522713602 207.1204403723443)"^^<http://www.opengis.net/ont/geosparql#wktLiteral>	kg:Parking_LumioseCity_68	"POLYGON ((386.2380121687817 207.5780813935397, 386.4322413815465 202.61630433497015, 375.44164361279786 202.07786128026584, 369.8615118820355 224.1679636349119, 386.2380121687817 207.5780813935397))"^^<http://www.opengis.net/ont/geosparql#wktLiteral>

3.

Query

```
PREFIX kg: <http://www.ai.di.uoa.gr/pokemon/>
PREFIX geof: <http://www.opengis.net/def/function/geosparql/>
SELECT ?location
        ?locWKT
        (COUNT(DISTINCT ?clue) AS ?numClues)
WHERE {
# the 5 parking locations (from C1)
VALUES ?location {
   kg:Parking_LumioseCity_21
   kg:Parking_LumioseCity_68
   kg:Parking_LumioseCity_108
   kg:Parking_LumioseCity_101
   kg:Parking_LumioseCity_100
  } # numerator proxy: geometry of the location (needed to compute area externally)
?location kg:hasLocation ?locWKT .
# denominator: number of clues whose POINT lies within the parking polygon
OPTIONAL {
?clue a kg:Clue ;
kg:hasLocation ?clueWKT .
FILTER( geof:sfWithin(?clueWKT, ?locWKT) )
}
}
GROUP BY ?location ?locWKT
ORDER BY ?location
```

Results

	location	locWKT	numClues
1	kg:Parking_LumioseCity_100	"POLYGON ((400.95800172457325 174.56028074617174, 385.1978246317435 182.8208843920269, 385.5626663159592 184.85017283350217, 399.8689372419413 186.3672989060587, 400.95800172457325 174.56028074617174))"^^<http://www.opengis.net/ont/geosparql#wktLiteral>	"1"^^xsd:integer
2	kg:Parking_LumioseCity_101	"POLYGON ((336.0268182753267 209.4222394888424, 335.9967620870416 217.0802761336244, 338.8291017821888 224.03615618610453, 344.0572080215613 220.3278864729608, 351.23703869655014 213.37030659960453, 336.0268182753267 209.4222394888424))"^^<http://www.opengis.net/ont/geosparql#wktLiteral>	"1"^^xsd:integer
3	kg:Parking_LumioseCity_108	"POLYGON ((377.703010326816 174.5376565516479, 376.68192741214864 174.20629993827137, 375.8388936333329 175.84285626455463, 377.85724313545256 176.0966510907535, 377.92225192929953 175.55114015767944, 377.703010326816 174.5376565516479))"^^<http://www.opengis.net/ont/geosparql#wktLiteral>	"2"^^xsd:integer
4	kg:Parking_LumioseCity_21	"POLYGON ((380.4228509911085 176.16431383682817, 376.46062110965346 183.00739674732844, 379.81939696694485 189.2600769215616, 382.27167670568696 192.74894390225745, 388.25739483991214 190.1869521694527, 394.5269161099987 184.0141956796327, 392.95848391629465 183.03065066851732, 380.4228509911085 176.16431383682817))"^^<http://www.opengis.net/ont/geosparql#wktLiteral>	"2"^^xsd:integer
5	kg:Parking_LumioseCity_68	"POLYGON ((386.2380121687817 207.5780813935397, 386.4322413815465 202.61630433497015, 375.44164361279786 202.07786128026584, 369.8615118820355 224.1679636349119, 386.2380121687817 207.5780813935397))"^^<http://www.opengis.net/ont/geosparql#wktLiteral>	"3"^^xsd:integer

The query did not return results because the GraphDB repository used for this assignment does not support an area function that can be evaluated inside SPARQL. In particular, call to geof:area(...) (GeoSPARQL function namespace) function fail with an "Unknown function …/area" evaluation error. Therefore, the square meters per clue value cannot be computed within the SPARQL engine, even though each parking location has a polygon geometry (kg:hasLocation) and clues can be spatially counted (e.g., with geof:sfWithin).

Consequently, the correct workflow is to (i) retrieve from SPARQL the parking geometry (WKT) and the number of clues per parking (see above), and then (ii) compute the polygon area in square meters externally (e.g., QGIS/PostGIS/Python) and finally derive:

$$\mathrm{m}^2/\mathrm{clue} \;=\; \frac{\mathrm{Area(location)}}{\#\mathrm{Clues\ found\ in\ that\ location}}$$

D)

Query

```
PREFIX kg:   <http://www.ai.di.uoa.gr/pokemon/>
PREFIX geof: <http://www.opengis.net/def/function/geosparql/>

SELECT DISTINCT ?town ?name
WHERE {
 kg:CentricoPlaza kg:hasLocation ?plazaWKT .
```

```
 ?town a kg:Town ;
     kg:hasName ?name ;
     kg:hasLocation ?townWKT .

 FILTER( STRSTARTS(STR(?name), "D") )

 BIND(
  geof:distance(?townWKT, ?plazaWKT)
  AS ?dist
 )
 FILTER( ?dist < 50)

 ?river a kg:River ;
      kg:hasLocation ?riverWKT .
 FILTER( geof:sfIntersects(?riverWKT, ?townWKT) )

 {
  SELECT ?town
  WHERE {
   ?town a kg:Town ; kg:hasLocation ?tWKT .
   ?forest a kg:Forest ; kg:hasLocation ?fWKT .
   FILTER( geof:sfTouches(?tWKT, ?fWKT) || geof:sfIntersects(?tWKT, ?fWKT) )
  }
  GROUP BY ?town
  HAVING (COUNT(DISTINCT ?forest) = 3)
 }
}
```

Results

The query is syntactically correct and methodologically appropriate, yet it returns **no results**. This outcome indicates that, within the scope of the available dataset, there is no town for which all the specified criteria are satisfied simultaneously.

More specifically, the conjunction of multiple constraints—namely a name starting with the letter D, proximity within 50 km, intersection with a river, and adjacency to exactly three forests—may legitimately result in an empty result set if such a combination does not exist in the data.

At the same time, this result may also reflect **limitations inherent to the dataset and the GeoSPARQL support**, particularly in the evaluation of complex spatial relations (e.g., river–polygon crossings or exact forest adjacency counts). As observed in other queries on this dataset, certain spatial relationships may not be fully detectable or consistently represented.

E)

Query

```
PREFIX kg:   <http://www.ai.di.uoa.gr/pokemon/>
PREFIX geof: <http://www.opengis.net/def/function/geosparql/>

SELECT DISTINCT ?pokemon ?name
WHERE {
 # Anne (current location)
 kg:Anne a kg:Person ;
       kg:hasLocation ?anneWKT .
```

```
# Pokemon with level >= 60
?pokemon a kg:Pokemon ;
        kg:hasName ?name ;
        kg:isLevel ?level ;
        kg:competedIn ?tournament .
FILTER(?level >= 60)

# Tournament took place in an Ice-type gym
?tournament a kg:Tournament ;
           kg:takesPlaceIn ?gym .
?gym a kg:Gym ;
    kg:hasGymType "Ice" ;
    kg:hasLocation ?gymWKT .

# Distance in km
BIND(
  geof:distance(?anneWKT, ?gymWKT)
  AS ?distDeg
)
FILTER(?distDeg > 80)
}
```

Results

	pokemon	name
1	kg:Machoke_7	"Machoke"
2	kg:Abra_33	"Abra"
3	kg:Dratini_185	"Dratini"
4	kg:Boldore_216	"Boldore"
5	kg:Dragonite_19	"Dragonite"
6	kg:Kakuna_74	"Kakuna"
7	kg:Nidoqueen_23	"Nidoqueen"
8	kg:Kakuna_45	"Kakuna"
9	kg:Beedrill_21	"Beedrill"
10	kg:Alakazam_79	"Alakazam"
11	kg:Wartortle_31	"Wartortle"
12	kg:Metapod_29	"Metapod"
13	kg:Ivysaur_41	"Ivysaur"
14	kg:Ivysaur_30	"Ivysaur"
15	kg:Vileplume_25	"Vileplume"
16	kg:Beedrill_34	"Beedrill"
17	kg:Golem_44	"Golem"
18	kg:Squirtle_71	"Squirtle"
19	kg:Abra_76	"Abra"
20	kg:Poliwrath_13	"Poliwrath"

F)

1.

Query

```
PREFIX kg:   <http://www.ai.di.uoa.gr/pokemon/>
PREFIX geof: <http://www.opengis.net/def/function/geosparql/>

SELECT ?anneWKT ?gym ?gymWKT ?dist
WHERE {
 kg:Anne kg:hasLocation ?anneWKT ;
         kg:leadsGym ?gym .

 ?gym a kg:Gym ;
      kg:hasLocation ?gymWKT .

 BIND(
  geof:distance(
    ?anneWKT,
    ?gymWKT
  ) AS ?dist
 )
}
```

Results

	anneWKT	gym	gymWKT	dist
1	"POINT (304.657277941236 62.20954883005388)"^^<http://www.opengis.net/ont/geosparql#wktLiteral>	kg:DriftveilCityGym	"POLYGON ((251.60005387215836 116.67768676720482, 252.8933299997541 115.15902789729662, 271.3426079250464 92.38816091365385, 247.64169188886592 88.85090891022419, 251.60005387215836 116.67768676720482))"^^<http://www.opengis.net/ont/geosparql#wktLiteral>	"44.95126100099979"^^xsd:double

2.

Query

```
PREFIX kg:   <http://www.ai.di.uoa.gr/pokemon/>
PREFIX geof: <http://www.opengis.net/def/function/geosparql/>

SELECT ?dAnneGym ?dPlazaPolice ?comparison
WHERE {
 # (A) Anne → her Gym
 kg:Anne kg:hasLocation ?anneWKT ;
        kg:leadsGym ?gym .
 ?gym kg:hasLocation ?gymWKT .

 BIND(
  geof:distance(?anneWKT, ?gymWKT)
  AS ?dAnneGym
 )

 # (B) Centrico Plaza → Police Department (Lumiose City)
 kg:CentricoPlaza kg:hasLocation ?plazaWKT .
 kg:PoliceStation_LumioseCity kg:hasLocation ?psWKT .
```

```
 BIND(
   geof:distance(?plazaWKT, ?psWKT)
   AS ?dPlazaPolice
 )

 # (C) Compare distances
 BIND(
   IF(
     !BOUND(?dAnneGym) || !BOUND(?dPlazaPolice),
     "comparison not possible (distance unbound)",
     IF(
       ?dAnneGym < ?dPlazaPolice, "Anne–Gym is shorter",
       IF(
         ?dAnneGym > ?dPlazaPolice, "Anne–Gym is longer",
         "equal"
       )
     )
   )
   AS ?comparison
 )
}
```

Results

	dAnneGym	dPlazaPolice	comparison
1	"44.95126100099979"^^xsd:double	"6.659202025466399"^^xsd:double	"Anne–Gym is longer"

G)

1.

Query

```
PREFIX geo:  <http://www.opengis.net/ont/geosparql#>
PREFIX geof: <http://www.opengis.net/def/function/geosparql/>

SELECT DISTINCT ?district ?name
WHERE {
 <http://yago-knowledge.org/resource/geoentity_City_of_Belfast_3333223>
   geo:hasGeometry/geo:asWKT ?bWKT .

 ?district
   <http://kr.di.uoa.gr/yago2geo/ontology/hasGADM_Description> "District" ;
   <http://kr.di.uoa.gr/yago2geo/ontology/hasGADM_Name> ?name ;
   geo:hasGeometry/geo:asWKT ?dWKT .

 FILTER( geof:sfIntersects(?dWKT, ?bWKT) )
 FILTER( ?district != <http://yago-knowledge.org/resource/geoentity_City_of_Belfast_3333223> )
}
```

Results

	district	name
1	http://kr.di.uoa.gr/yago2geo/resource/gadmentity_North_Down_and_Ards_GBR.2.11_1	"North Down and Ards"@en
2	http://yago-knowledge.org/resource/geoentity_Antrim_and_Newtownabbey_11353069	"Antrim and Newtownabbey"@en
3	http://yago-knowledge.org/resource/geoentity_Lisburn_and_Castlereagh_11353072	"Lisburn and Castlereagh"@en

2.

Query

```
PREFIX geo:  <http://www.opengis.net/ont/geosparql#>
PREFIX geof: <http://www.opengis.net/def/function/geosparql/>
PREFIX yago: <http://kr.di.uoa.gr/yago2geo/ontology/>

SELECT ?river ?riverName (COUNT(DISTINCT ?district) AS ?numDistricts)
WHERE {
  # districts in Northern Ireland (as in your Belfast query)
  ?district yago:hasGADM_Description "District" ;
           yago:hasGADM_Name ?districtName ;
           geo:hasGeometry/geo:asWKT ?dWKT .

  # rivers with hasLocation (LINESTRING WKT)
  ?river a ?t ;
         <http://www.ai.di.uoa.gr/pokemon/hasLocation> ?rWKT .
  FILTER(CONTAINS(LCASE(STR(?t)), "river"))

  OPTIONAL { ?river <http://www.ai.di.uoa.gr/pokemon/hasName> ?riverName }

  # river crosses a district
  FILTER( geof:sfIntersects(?rWKT, ?dWKT) )
}
GROUP BY ?river ?riverName
HAVING (COUNT(DISTINCT ?district) >= 2)
ORDER BY DESC(?numDistricts) LCASE(STR(?riverName))
LIMIT 50
```

Results

river	riverName	numDistricts
No data available in table		

The query does not return any results, not due to an error in its formulation, but because of limitations in the available data and the spatial processing capabilities of the system.

First, although the query correctly searches for rivers that spatially intersect with at least two districts of Northern Ireland, the river geometries originate from the Pokémon knowledge graph, while the district geometries belong to the Northern Ireland (YAGO2Geo) dataset. These datasets describe different spatial domains and are defined in incompatible or unrelated coordinate spaces. As a result, spatial relationships between river and district geometries cannot be meaningfully evaluated as true.

Second, even if the geometries were conceptually related, GraphDB's GeoSPARQL implementation has known limitations when evaluating topological predicates such as geof:sfIntersects for certain geometry combinations, particularly **LINESTRING–POLYGO**N. In such cases, the function may fail to return true even when a river geometrically crosses a district, due to implementation and precision constraints of the spatial engine.

Consequently, no river can be identified as crossing at least two districts of Northern Ireland based on the provided datasets. The absence of results therefore reflects data interoperability limitations and GeoSPARQL processing constraints, rather than a flaw in the query or the analytical approach.

3.

Query

```
PREFIX geo:  <http://www.opengis.net/ont/geosparql#>
PREFIX geof: <http://www.opengis.net/def/function/geosparql/>

SELECT ?bboxWKT
WHERE {
 <http://yago-knowledge.org/resource/Northern_Ireland>
   geo:hasGeometry/geo:asWKT ?niWKT .

 BIND( geof:envelope(?niWKT) AS ?bboxWKT )
}
```

Results

	bboxWKT
1	"POLYGON ((-8.17156887 54.02264023, -8.17156887 55.31263733, -5.43249989 55.31263733, -5.43249989 54.02264023, -8.17156887 54.02264023))" ^^<http://www.opengis.net/ont/geosparql#wktLiteral>
2	"POLYGON ((-8.17750233546595 54.0227239201828, -8.17750233546595 55.3129838549032, -5.43278930668693 55.3129838549032, -5.43278930668693 54.0227239201828, -8.17750233546595 54.0227239201828))"^^<http://www.opengis.net/ont/geosparql#wktLiteral>

4.

Query

```
PREFIX geo:  <http://www.opengis.net/ont/geosparql#>
PREFIX geof: <http://www.opengis.net/def/function/geosparql/>
PREFIX yago: <http://kr.di.uoa.gr/yago2geo/ontology/>

SELECT ?d1 ?name (COUNT(DISTINCT ?d2) AS ?nNeighbors)
WHERE {
 ?d1 yago:hasGADM_Description "District" ;
     yago:hasGADM_Name ?name ;
     geo:hasGeometry/geo:asWKT ?wkt1 .

 ?d2 yago:hasGADM_Description "District" ;
     geo:hasGeometry/geo:asWKT ?wkt2 .
```

```
 FILTER(?d1 != ?d2)
 FILTER( geof:sfTouches(?wkt1, ?wkt2) || geof:sfIntersects(?wkt1, ?wkt2) )
}
GROUP BY ?d1 ?name
ORDER BY DESC(?nNeighbors)
```

Results

	d1	name	nNeighbors
1	http://yago-knowledge.org/resource/Mid-Ulster_District	"Mid Ulster"@en	"6"^^xsd:integer
2	http://yago-knowledge.org/resource/geoentity_Antrim_and_Newtownabbey_11353069	"Antrim and Newtownabbey"@en	"5"^^xsd:integer
3	http://yago-knowledge.org/resource/geoentity_Lisburn_and_Castlereagh_11353072	"Lisburn and Castlereagh"@en	"5"^^xsd:integer
4	http://yago-knowledge.org/resource/geoentity_Armagh_City_Banbridge_and_Craigavon_11353076	"Armagh, Banbridge and Craigavon"@en	"4"^^xsd:integer
5	http://kr.di.uoa.gr/yago2geo/resource/gadmentity_North_Down_and_Ards_GBR.2.11_1	"North Down and Ards"@en	"3"^^xsd:integer
6	http://yago-knowledge.org/resource/geoentity_Newry_Mourne_and_Down_11353077	"Newry, Mourne and Down"@en	"3"^^xsd:integer
7	http://yago-knowledge.org/resource/geoentity_Derry_City_and_Strabane_11353073	"Derry and Strabane"@en	"3"^^xsd:integer
8	http://yago-knowledge.org/resource/geoentity_Derry_City_and_Strabane_11353073	"<Null>"	"3"^^xsd:integer
9	http://yago-knowledge.org/resource/geoentity_Causeway_Coast_and_Glens_11353070	"Causeway Coast and Glens"@en	"3"^^xsd:integer
10	http://yago-knowledge.org/resource/geoentity_Mid_and_East_Antrim_11353078	"Mid and East Antrim"@en	"3"^^xsd:integer
11	http://yago-knowledge.org/resource/geoentity_City_of_Belfast_3333223	"Belfast"@en	"3"^^xsd:integer
12	http://yago-knowledge.org/resource/geoentity_Fermanagh_and_Omagh_11353071	"Fermanagh and Omagh"@en	"2"^^xsd:integer

Based on the spatial adjacency analysis of Northern Ireland districts (using polygon–polygon topological relations), **Mid Ulster** is the district with the highest number of neighbouring districts.

As shown in the query results, **Mid Ulster has 6 neighbouring districts**, which is the maximum among all districts in Northern Ireland. This conclusion follows directly from counting the number of distinct districts whose geometries touch the geometry of each district, and then ranking the results by this count.

5.

Query

```
PREFIX geo:  <http://www.opengis.net/ont/geosparql#>
PREFIX geof: <http://www.opengis.net/def/function/geosparql/>
PREFIX rdfs: <http://www.w3.org/2000/01/rdf-schema#>
PREFIX yago: <http://kr.di.uoa.gr/yago2geo/ontology/>

SELECT DISTINCT ?district ?districtName ?lake ?lakeLabel ?lakeArea
WHERE {
 # --- 2nd largest lake in Northern Ireland (by hasArea) ---
 {
  SELECT ?lake ?lakeLabel ?lakeArea ?lakeWKT
  WHERE {
   # Northern Ireland geometry
   <http://yago-knowledge.org/resource/Northern_Ireland>
    geo:hasGeometry/geo:asWKT ?niWKT .

   # Lake with label + geometry + stored area
   ?lake rdfs:label ?lakeLabel ;
       geo:hasGeometry/geo:asWKT ?lakeWKT ;
       yago:hasArea ?lakeArea .

   FILTER(
    CONTAINS(LCASE(STR(?lakeLabel)), "lough") ||
    CONTAINS(LCASE(STR(?lakeLabel)), "lake")
   )

   # ensure the lake is in Northern Ireland
   FILTER( geof:sfIntersects(?lakeWKT, ?niWKT) )
  }
  ORDER BY DESC(?lakeArea)
  OFFSET 1
  LIMIT 1
 }

 # --- Districts intersecting that lake ---
 ?district yago:hasGADM_Description "District" ;
       yago:hasGADM_Name ?districtName ;
       geo:hasGeometry/geo:asWKT ?distWKT .

 FILTER( geof:sfIntersects(?distWKT, ?lakeWKT) )
}
ORDER BY LCASE(STR(?districtName))
```

Results

	district	districtName	lake	lakeLabel	lakeArea
1	http://yago-knowledge.org/resource/geoentity_Antrim_and_Newtownabbey_11353069	"Antrim and Newtownabbey"@en	http://yago-knowledge.org/resource/Lough_Neagh	"Lough Neagh"@cy	"0.05294701270643011"^^xsd:double
2	http://yago-knowledge.org/resource/geoentity_Armagh_City_Banbridge_and_Craigavon_11353076	"Armagh, Banbridge and Craigavon"@en	http://yago-knowledge.org/resource/Lough_Neagh	"Lough Neagh"@cy	"0.05294701270643011"^^xsd:double
3	http://yago-knowledge.org/resource/geoentity_Lisburn_and_Castlereagh_11353072	"Lisburn and Castlereagh"@en	http://yago-knowledge.org/resource/Lough_Neagh	"Lough Neagh"@cy	"0.05294701270643011"^^xsd:double
4	http://yago-knowledge.org/resource/Mid-Ulster_District	"Mid Ulster"@en	http://yago-knowledge.org/resource/Lough_Neagh	"Lough Neagh"@cy	"0.05294701270643011"^^xsd:double

6.

Query

```
PREFIX yago: <http://kr.di.uoa.gr/yago2geo/ontology/>

SELECT ?district ?name ?area ?population
WHERE {
 {
  SELECT ?district ?name ?area
  WHERE {
   ?district yago:hasGADM_Description "District" ;
        yago:hasGADM_Name ?name ;
        yago:hasArea ?area .
  }
  ORDER BY DESC(?area)
  LIMIT 3
 }

 ?district yago:hasPopulation ?population .
}
ORDER BY DESC(?population)
LIMIT 1
```

Results

	district	name	area	population
1	http://yago-knowledge.org/resource/geoentity_Causeway_Coast_and_Glens_11353070	"Causeway Coast and Glens"@en	"0.2839998039473298"^^xsd:double	"141954"^^xsd:integer